
LISTING PROGRAM
' Gambas class file

PUBLIC $Koneksi AS Connection

PUBLIC $inData AS Result

PUBLIC $outData AS Result

PUBLIC CONST pid AS String = "No Proses : "

'////////////// fungsi funsi //////////////////////////////

PUBLIC SUB Aktif(status AS Integer)

DIM i AS Integer

 IF status = 0 THEN

   FOR i = 1 TO 4

    tabs[i].Enabled = FALSE

   NEXT

 ELSE

   FOR i = 1 TO 4

    tabs[i].Enabled = TRUE

   NEXT

 ENDIF

END

PUBLIC SUB Konek()

 DIM NamaDB AS String

 $koneksi = NEW Connection

 TRY $koneksi.Close

 namadb = txdb.Text

 WITH $koneksi

    .Type = "mysql"

    .Host = txHost.Text

    .Login = txUser.Text

    .Password = txPassword.Text

 END WITH

 $koneksi.Name = Namadb

 $koneksi.Open

 SHELL "gnokii --identify > detail-hp" WAIT '& detail WAIT

 txdetail.Text = file.Load("detail-hp")

 IF txpassword.Text <> "" THEN

  SHELL "smsd -m mysql -d " & namadb & " -u " & txuser.Text & " -p " & txpassword.Text & "&"

 ELSE

  SHELL "smsd -m mysql -d " & namadb & " -u " & txuser.Text & "&"

 ENDIF

 aktif(1)

 SHELL "ps -C smsd o pid > id | ps -C smsd o pid > id" WAIT

 SHELL "tail -n 1 id > pid-smsd | tail -n 1 id > pid-smsd" WAIT

 lbpid.Text = pid & file.Load("pid-smsd") 'pid 'file.Load("pid-smsd")

CATCH

  Message.Error(DConv(Error.Text))

END

PUBLIC SUB DisKonek()

 TRY $koneksi.Close

'/////// kill sms daemon /////////////////

'  SHELL "kill " & txpid.Text & " " & txpid.Text

'  atau bisa juga dengan

'   SHELL "kill $(ps -C smsd o pid)"

  SHELL "skill -KILL -f -c smsd "

'//// hapus file

 WAIT 2

 txdetail.Clear

 lbpid.Text = pid

 btkonek.Text = "Connect"

 aktif(0)

END

PUBLIC SUB TampilkanTabel(tblView AS Control, queri AS String)

 DIM isidata AS ResultField

 DIM i AS Integer

 $outdata = $koneksi.Exec(queri)

 WITH tblview

  .Rows.Count = 0

  .Columns.Count = $outdata.Fields.Count

  FOR EACH isidata IN $outdata.Fields

   .Columns[i].Text = isidata.Name

   INC i

  NEXT

 .Rows.Count = $outdata.Count

 .Row = 0

 END WITH

CATCH

  Message.Error("Gagal mengambil data." & "\n\n" & Error.Text)

END

PUBLIC SUB HapusData(Tabel AS String, Kriteria AS String)

 DIM queri AS String

 DIM hapus AS Result

 queri = "delete from " & tabel & " where " & kriteria

 hapus = $koneksi.Exec(queri)

END

PUBLIC SUB TambahData(Tabel AS String, Field AS String, Nilai AS String)

 DIM queri AS String

 DIM Tambah AS Result

 queri = "insert into " & tabel & "(" & field & ")" & " values('" & nilai & "')" 

 tambah = $koneksi.Exec(queri)

END

'////////////////////// form utama ////////////////////////

PUBLIC SUB Form_Open()

 aktif(0)

END

PUBLIC SUB tabS_Click()

 SELECT CASE tabs.Index

 CASE 2

  lihatsms

 CASE 3

  lihatkunci

 CASE 4

  lihathasil

 END SELECT

END

'////////////////////////////// menu //////////////////////////////

PUBLIC SUB mnexit_Click()

 TRY DisKonek 'WAIT

'  WAIT 3

 QUIT

END

'//////////////////// bagian informasi ///////////////////////////////

PUBLIC SUB btKonek_Click()

 IF btkonek.Text = "Connect" THEN

  Konek

  btkonek.Text = "Disconnect"

 ELSE

  diskonek

 ENDIF

END

'///////////////////// bagian kirim sms /////////////////////////////////

PUBLIC SUB btKirim_Click()

 'KirimSMS(txnotujuan.Text, txisi.Text)

 tambahdata("outbox", "number,text", txnotujuan.Text & "','" & txisi.Text)

 message.Info("Pesan SMS anda sedang dikirimkan....")

END

PUBLIC SUB txIsi_KeyPress()

 sisa.Value = 159 - Len(txisi.Text)

 IF sisa.Value <= 0 THEN message.Warning("Quota sms anda telah habis")

END

PUBLIC SUB btReset_Click()

 txisi.Clear

END

'///////////////////////bagian baca sms ///////////////////////////////////

PUBLIC SUB LihatSMS()

 tampilkantabel(tvdatasms, "SELECT id,number,smsdate,text FROM inbox ORDER BY id DESC")

 WITH tvdatasms

  .Columns[0].Width = 40

  .Columns[1].Width = 100

  .Columns[2].Width = 110

  .Columns[3].Width = 330

 END WITH

 tvDataSMS_Click()

END

PUBLIC SUB tvdatasms_Data(Row AS Integer, Column AS Integer)

 $outdata.MoveTo(Row)

 tvdatasms.Data.Text = Str($outdata[tvdatasms.Columns[Column].Text])

END

PUBLIC SUB btBalas_Click()

 tabs.Index = 1

 txnotujuan.Text = txnopengirim.Text

 txisi.SetFocus

END

PUBLIC SUB tvDataSMS_Click()

 txnopengirim.Text = tvdatasms[tvdatasms.Row, 1].Text

 txpesan.Text = tvdatasms[tvdatasms.Row, 3].Text

 frtgl.Text = "| Diterima pada : " & tvdatasms[tvdatasms.Row, 2].Text & " |"

END

PUBLIC SUB btHapus_Click()

 hapusdata("inbox", "id=" & tvdatasms[tvdatasms.Row, 0].Text & " and number='" & txnopengirim.Text & "'")

 btreload_click()

END

PUBLIC SUB btReload_Click()

 lihatsms

END

'//////////////////////////////bagian isi polling ////////////////////

PUBLIC SUB LihatKunci()

 tampilkantabel(tvdatapolling, "select id,kunci,pilihan from polling")

  WITH tvdatapolling

  .Columns[0].Width = 40

  .Columns[1].Width = 100

  .Columns[2].Width = 110

 END WITH

END

PUBLIC SUB btTambah_Click()

 tambahdata("polling", "kunci,pilihan", txkunci.Text & "','" & txpilihan.Text)

 lihatkunci

END

PUBLIC SUB tvDataPolling_Data(Row AS Integer, Column AS Integer)

 $outdata.MoveTo(Row)

 tvdatapolling.Data.Text = Str($outdata[tvdatapolling.Columns[Column].Text])

END

PUBLIC SUB btHapusKunci_Click()

 hapusdata("polling", "kunci='" & tvdatapolling[tvdatapolling.Row, 1].Text & "' and pilihan='" & tvdatapolling[tvdatapolling.Row, 2].Text & "'")

 lihatkunci

END

'/////////////////////////////bagian Hasil polling ////////////////////////////

PUBLIC SUB HitungHasil()

DIM i AS Integer

DIM nKunci AS Integer

DIM nPil AS Integer

DIM oPil AS Integer

DIM Kunci AS String

DIM Pilihan AS String

DIM queri1 AS String

DIM queri2 AS String

DIM Hasil AS String

 FOR i = 0 TO tvhasilpolling.Rows.Count - 1

  tvhasilpolling.Row = i

  kunci = tvhasilpolling[i, 1].Text

  pilihan = tvhasilpolling[i, 2].Text

  nkunci = Len(kunci)

  opil = nkunci + 2

  npil = Len(pilihan)

  queri1 = "select count(text) as jumlah from inbox where left(text," & nkunci & ")='" & kunci & "' AND Mid(text," & opil & "," & npil & ")='" & pilihan & "'"

  $indata = $koneksi.Exec(queri1)

  hasil = $indata!jumlah

  queri2 = "UPDATE polling SET hasil='" & hasil & "' WHERE kunci='" & kunci & "' and pilihan='" & pilihan & "'"

  $indata = $koneksi.Exec(queri2)

 NEXT

 'text1.Text = kunci & "/" & pilihan & "/" & nkunci & "/" & npil & "/" & opil & "/" & hasil

END

PUBLIC SUB LihatHasil()

 tampilkantabel(tvhasilpolling, "select id,kunci,pilihan,hasil from polling")

  WITH tvhasilpolling

  .Columns[0].Width = 40

  .Columns[1].Width = 100

  .Columns[2].Width = 110

  .Columns[3].Width = 110

 END WITH

END

PUBLIC SUB tvhasilpolling_Data(Row AS Integer, Column AS Integer)

 $outdata.MoveTo(Row)

 tvhasilpolling.Data.Text = Str($outdata[tvhasilpolling.Columns[Column].Text])

END

PUBLIC SUB btRePoll_Click()

 hitunghasil

 lihathasil

END

MANUAL PROGRAM

1. Menjalankan program

b. Masukkan nama atau lokasi database server pada host. Lokasi dapat berupa nama server maupun alamat IP komputer yang dituju.
c. Ketikkan nama database yang digunakan pada kolom database.

d. Masukkan nama user dan password untuk database tersebut.

e. Klik tombol Connect untuk menjalankan program, sehingga pada kotak detail akan tampil informasi detail ponsel yang digunakan.

2. Mengirim Sms

a. Klik pada tab strip kirim sms untuk membuka jendela pengiriman sms.

b. Masukkan terlebih dahulu no handphone tujuan sms akan dikirimkan, kemudian ketikkan isi pesan pada kotak yang tersedia.

c. Untuk mengirimkan sms yang telah diketikkan dilakukan dengan cara menekan tombol kirim sms.

d. Tekan tombol reset bila ingin menghapus atau membatalkan isi pesan yang akan dikirimkan.

3. Membaca Sms

a. Klik pada tab strip baca sms untuk membuka jendela baca sms. Pada layar akan langsung ditampilkan sms terakhir yang baru masuk.

b. Klik pada salah satu baris pada tabel daftar sms masuk, untuk menampilkan isi pesan yang masuk dengan lebih detail.

c. Untuk membalas sms yang sedang dibuka secara langsung dilakukan dengan menekan tombol balas sms.

d. Untuk menghapus pesan dilakukan dengan cara klik salah satu pesan yang ingin dihapus, kemudian klik pada tombol hapus sms.

4. Membuat Aturan Polling

a. Masukkan kata kunci yang akan dijadikan kunci sebuah polling sms pada kotak yang tersedia.

b. Masukkan kata pilihan untuk kata kunci tersebut pada kotak.

c. Klik tambah kunci sms untuk menambahkan kata kunci ini kedalam daftar kata kunci polling.

d. Ulangi langkah diatas untuk menambah kata kunci atau kata pilihan yang lain.

e. Untuk menghapus kata kunci dari daftar dilakukan dengan cara klik kata kunci yang diinginkan kemudian klik pada hapus kata kunci.

5. Menghitung Hasil Polling

a. Klik pada tab strip hasil polling untuk melihat hasil perolehan polling sms.

b. Klik tombol Reload untuk mengupdate hasil yang terbaru.
6. Memutus koneksi
memutus kerja disini adalah memutus hubungan dengan sms gateway yang sedang berjalan dalam sistem. Dilakukan dengan cara menekan tombol Disconnect sehingga tampilan tombol akan kembali ke bentuk Connect dan menjadi tidak aktifnya tab strip.

7. Keluar dari program utama.

Untuk keluar dari program atau menutup program dilakukan dengan cara memilih menu File | Exit.























































































































































































































































































